

Hewlett-Packard

HP-19C/HP-29C SOLUTIONS

STUDENT ENGINEERING

INTRODUCTION

This HP-19C/HP-29C Solutions book was written to help you get the most from your calculator. The programs were chosen to provide useful calculations for many of the common problems encountered.

They will provide you with immediate capabilities in your everyday calculations and you will find them useful as guides to programming techniques for writing your own customized software. The comments on each program listing describe the approach used to reach the solution and help you follow the programmer's logic as you become an expert on your HP calculator.

You will find general information on how to key in and run programs under "A Word about Program Usage" in the Applications book you received with your calculator.

We hope that this Solutions book will be a valuable tool in your work and would appreciate your comments about it.

TABLE OF CONTENTS

RESISTIVE/REACTIVE CIRCUIT CALCULATIONS

This program performs resonance calculations for R-L-C circuits, calculates the reactance of inductive and capacitive branches, the equivalent value of series capacitors or parallel resistors and inductors, and performs power calculations for resistive branches using straightforward manipulations of the following equations:

$$f_r = \frac{1}{2\pi\sqrt{LC}}$$

$$X_C = \frac{1}{2\pi fC}$$

$$X_L = 2\pi fL$$

$$P = I^2R = E^2/R$$

$$\frac{A_1A_2}{A_1+A_2} = A_3$$

where

f_r = resonant frequency in hertz

L = inductance in henrys

C = capacitance in farads

X_C = capacitive reactance in Ω

X_L = inductive reactance in Ω

P = power in watts

I = current in amps

R = resistance in Ω

E = voltage in volts

A_1, A_2 = the values of two parallel resistors in ohms, two parallel inductors in henrys, or two series capacitors in farads

A_3 = the resultant, equivalent resistance in ohms, inductance in henrys, or capacitance in farads

NOTE: Given a resistance or capacitance, A_1, the value of the circuit element required to produce a desired resultant resistance or capacitance may be calculated by entering A_1 as a negative value.

EXAMPLES:

1. C = .01μF, L = 160μh. Calculate f_r
2. L = 2.5h, f_r = 60H_z Calculate C and X_L at f_r
3. E = 345v, R = 1.25MΩ Calculate P and I
4. R_1= 120Ω, R_2 = 240Ω
   a. Find the equivalent resistance of these two resistors in parallel,R_3.
   b. Parallel R_3 with 50Ω.
   c. Find the resistance required for a resultant resistance of 25Ω.

SOLUTION:

```
             ENG4
   160.-06 ENT↑
   0.01-06 GSB1
 125.82+03  ***     f_r

   60.0000 ENT↑
    2.5000 GSB2
 2.8145-06  ***     C

   60.0000 ENT↑
    2.5000 GSB4
 942.48+00  ***     X_L

  345.0000 ENT↑
   1.25+06 GSB5
 95.220-03  ***     P

   1.25+06 GSB7
 276.00-06  ***     I

  120.0000 ENT↑
  240.0000 GSB9
 80.000+00  ***     R_3

   50.0000 GSB9     4b
 30.769+00  ***
             CHS
   25.0000 GSB9     4c
 133.33+00  ***
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program and choose an appropriate				
	display format				
2.	Calculate f_r	L	ENT		
		C	GSB	1	f_r
3.	Calculate L or C	f_r	ENT		
		C or L	GSB	2	L or C
	REACTANCE CALCULATIONS				
4.	Calculate X_C	f	ENT		
		C	GSB	3	X_C
5.	Calculate X_L	f	ENT		
		L	GSB	4	X_L
	POWER CALCULATIONS				
6a	Given E: Calculate P	E	ENT		
		R	GSB	5	P
6b	Given I: Calculate P	I	ENT		
		R	GSB	6	P
	Given P,R:				
7a	Calculate I	P	ENT		
		R	GSB	7	I
7b	Calculate E	P	ENT		
		R	GSB	8	E
	PARALLEL RESISTANCE/SERIES CAPACITANCE				
8.	Given two circuit elements, calculate resultant	A_1	ENT		
		A_2	GSB	9	A_3
9.	Given one circuit element and the desired	A_1	CHS	ENT	
	resultant value, calculate the value of the	A_3	GSB	9	A_2
	circuit element required.				

Program Listings

01	*LBL1	
02	×	
03	√X	
04	GTO0	
05	*LBL2	
06	X⇄Y	
07	GSB0	
08	X²	
09	X⇄Y	
10	÷	
11	R/S	L or C
12	*LBL3	
13	×	
14	GTO0	
15	*LBL4	
16	×	
17	GSB0	
18	1/X	
19	R/S	X_L
20	*LBL5	
21	1/X	
22	*LBL6	
23	X⇄Y	
24	X²	
25	×	
26	R/S	P
27	*LBL7	
28	1/X	
29	*LBL8	
30	×	
31	√X	I or E
32	R/S	
33	*LBL0	
34	2	
35	×	
36	Pi	
37	×	
38	1/X	f_r or X_C
39	RTN	
40	*LBL9	
41	×	
42	ENT↑	
43	ENT↑	
44	LSTX	
45	÷	
46	LSTX	
47	+	A_3 or A_2
48	÷	
49	R/S	

REGISTERS

0	1	2	3	4	5
6	7	8	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

IMPEDANCE OF A LADDER NETWORK

This program computes the input impedance of an arbitrary ladder network. Elements are added one at a time starting from the right. The first element must be in parallel.

Suppose we have a network whose input admittance is Y_{in}. Adding a shunt R, L, or C, the input admittance becomes

$$Y_{new} = \begin{cases} Y_{in} + \left(\frac{1}{R} + j0\right) \\ Y_{in} + \left(0 - j\frac{1}{\omega L_p}\right) \\ Y_{in} + (0 + j\,\omega\, C_p) \end{cases}$$

Adding a series R, L, or C, we have

$$Y_{new} = \begin{cases} \left(\frac{1}{Y_{in}} + (R_s + j0)\right)^{-1} \\ \left(\frac{1}{Y_{in}} + (0 + j\,\omega\, L_s)\right)^{-1} \\ \left(\frac{1}{Y_{in}} + \left(0 - j\frac{1}{\omega C_s}\right)\right)^{-1} \end{cases}$$

The program converts this admittance to an impedance for display.

NOTE: An erroneous entry may be corrected by entering the negative of the incorrect value.

EXAMPLE:

f = 4 MHz



SOLUTION:

	FIX2	
4.+06	GSB1	
50.00	GSB3	$\|Z_{in}\|$
50.00	***	
	X⇄Y	
0.00	***	$\angle Z_{in}$
2400.-12	GSB5	
15.74	***	$\|Z_{in}\|$
	X⇄Y	
-71.66	***	$\angle Z_{in}$
2.56-06	GSB2	
	GSB5	
49.65	***	$\|Z_{in}\|$
	X⇄Y	
84.28	***	$\angle Z_{in}$
796.-12	GSB5	
497.69	***	$\|Z_{in}\|$
	X⇄Y	
0.99	***	$\angle Z_{in}$

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program and choose an appropriate display format.				
2.	Initialize	f	GSB	1	W
3.	Input a parallel element:				
		R_p	GSB	3	$\|Z_{in}\|$
			x↔y		$\angle Z_{in}$
		C_p	GSB	5	$\|Z_{in}\|$
			x↔y		$\angle Z_{in}$
		L_p	GSB	4	$\|Z_{in}\|$
			x↔y		$\angle Z_{in}$
4a	For a series element:				
		R_s	GSB	2	
			GSB	3	$\|Z_{in}\|$
			x↔y		$\angle Z_{in}$
		C_s	GSB	2	
			GSB	4	$\|Z_{in}\|$
			x↔y		$\angle Z_{in}$
		L_s	GSB	2	
			GSB	5	$\|Z_{in}\|$
			x↔y		$\angle Z_{in}$
4b	For a parallel element: Go to step 3				
5.	Repeat step 4 until all elements are entered.				

Program Listings

*LBL1	f
2	
x	
Pi	
x	Clear flag
CLRG	w
STO3	
R/S	
*LBL0	Z↔Y
R↓	
→P	
1/X	
X⇄Y	
CHS	
X⇄Y	
→R	
0	
RTN	R,C,L
*LBL2	Set flag (series)
STO0	
R/S	
*LBL3	
1/X	
RCL0	
X=0?	
GTO8	0,Y (parallel)
0	
GTO8	
*LBL4	0,Z (series)
RCL3	
x	
1/X	X_C or B_L
CHS	
0	
X⇄Y	
GTO8	
*LBL5	
RCL3	
x	X_L or B_C
0	
X⇄Y	
*LBL8	
STO7	
X⇄Y	
STO6	
RCL2	
RCL1	
RCL0	
X≠0?	
GSB0	Convert $Y_{in} \to Z_{in}$
R↓	
RCL6	
+	Add admittances
X⇄Y	or impedances
RCL7	
+	
X⇄Y	
RCL0	
X≠0?	
GSB0	
R↓	Convert Z→Y
STO1	
X⇄Y	
STO2	
X⇄Y	
0	
GSB0	Convert Y→Z
STO0	Clear flag
R↓	
→P	
R/S	*** $\lvert Z_{in} \rvert \angle Z_{in}$

*** "Printx" may be inserted.

REGISTERS

0 flag	1 $Re[Y_{in}]$	2 $Im[Y_{in}]$	3 $\omega = 2\pi f$	4	5
6 used	7 used	8	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

STANDARD RESISTANCE VALUES*

For a given tolerance, a "step size" is computed which is used to determine two values, one below and one above the non-standard resistance. These are converted by a subroutine to standard values, and the geometric mean of the latter is calculated. If the given non-standard value is below the mean then the lower standard value is selected; otherwise the larger value is selected.

NOTE: Incorrect results will be obtained for tolerances other than 5%, 10%, or 20%.

REFERENCE: International Telephone and Telegraph Corp. Reference Data for Radio Engineers, fourth edition, p. 78.

EXAMPLES: Find the closest standard values for the following:

$R_1 = 432\Omega$

$R_2 = 114\ K\Omega$

$R_3 = 3.5\ M\Omega$

SOLUTION:

```
            ENG4
    5.0000  GSB1    5%
  432.0000  GSB2
  430.00+00  ***    R'1

   114.+03  GSB2
  110.00+03  ***    R'2

    3.5+06  GSB2
 3.6000+06   ***    R'3

   10.0000  GSB1    10%
  432.0000  GSB2
  470.00+00  ***    R'1

   114.+03  GSB2
  120.00+03  ***    R'2

    3.5+06  GSB2
 3.3000+06   ***    R'3

   20.0000  GSB1    20%
  432.0000  GSB2
  470.00+00  ***    R'1

   114.+03  GSB2
  100.00+03  ***    R'2

    3.5+06  GSB2
 3.3000+06   ***    R'3
```

* Adapted from HP-65 Users' Library program #00915A by Jacob Jacobs.

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program and choose an appropriate display format				
2.	Enter tolerance (5,10, or 20)	T%	GSB	1	
3.	Enter non-standard resistor size and compute nearest standard value	R,Ω	GSB	2	R',Ω
4.	Change tolerance at any time by going to step 2				

Program Listings

01	*LBL1	
02	1	
03	2	
04	0	
05	÷	
06	10ˣ	
07	STO2	$10^{tol/120}$
08	R/S	
09	*LBL2	
10	LOG	
11	ENT↑	
12	INT	
13	STO4	
14	-	
15	1	
16	+	
17	10ˣ	
18	STO3	
19	1	
20	ST-4	
21	1	
22	0	
23	STO5	
24	*LBL0	
25	RCL3	
26	RCL5	
27	X>Y?	This step > Normal R?
28	GTO9	
29	RCL2	
30	x	
31	STO5	
32	GTO0	
33	*LBL9	
34	GSB8	
35	STO7	This step
36	RCL5	
37	RCL2	
38	÷	
39	GSB8	
40	STO6	Last step
41	RCL7	
42	x	
43	√X	$\sqrt{(\text{This step})*(\text{Last step})}$
44	RCL3	
45	X≤Y?	
46	GSB7	
47	RCL7	
48	RCL4	
49	10ˣ	
50	x	10 EXP R_7
51	R/S	*** R
52	*LBL8	Finds standard R value from multiple of step size
53	.	
54	5	
55	+	
56	INT	Round up
57	STO6	
58	2	
59	6	
60	X>Y?	
61	GTO6	
62	4	
63	7	
64	RCL6	
65	X>Y?	
66	GTO3	
67	1	26<R<47 then add 1
68	+	
69	RTN	
70	*LBL3	
71	8	
72	3	
73	RCL6	
74	X≠Y?	
75	RTN	
76	8	
77	2	
78	RTN	
79	*LBL7	
80	RCL6	
81	STO7	
82	RTN	
83	*LBL6	
84	RCL6	
85	RTN	
86	R/S	

*** "Printx" may be inserted before "R/S".

REGISTERS

0	1	2 Step size	3 Normal R	4 Exp of R	5 This step
6 Temp	7 Temp	8	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

EXPONENTIAL GROWTH OR DECAY

Many growth or decay phenomena encountered in science and engineering obey an exponential law of the general form:

$$X_t = X_{ss} - (X_{ss} - X_0)\, e^{\frac{-t}{\tau}}$$

where:

- X_t = Value at any time, t, (i.e., the instantaneous value)
- X_{ss} = Steady state value (i.e., at $t = \infty$)
- X_0 = Initial value (i.e., at t = 0)
- t = Elapsed time (time after t = 0)
- τ = Exponential time-constant for specific phenomena

This program provides interchangeable solutions for any one of the four variables X_t, X_{ss}, X_0 or t provided three variables and τ are known.

EXAMPLE 1:

Given a 5μF capacitor in series with a 1 megohm resistor. 1.5 seconds after the circuit is completed 125 volts are measured across R. To what voltage was the capacitor originally charged?

Note:

τ = the RC time-constant, and the voltage at $t = \infty$ is zero

SOLUTION:

```
  5.-06 ENT↑
  1.+06   x
        STO4  τ = time-constant
 125.00 STO1  v
   0.00 STO2  Xss
   1.50 STO3  time
        GSB0
 168.73  ***  volts
```

EXAMPLE 2:

A cobalt 60 source (half-life = 5.26 years) had an activity of 3.54 curies when purchased 8.5 years ago. What is its present activity?

Note:

Activity at $t = \infty$ will be zero, τ = half-life/LN2

SOLUTION:

```
  5.26 ENT↑
  2.00   LN
          ÷
       STO4  τ
  3.54 STO0  X0
  0.00 STO2  Xss
  8.50 STO3  t
       GSB1
  1.15  ***  curies
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1	Key in the program				
2	Store time constant	τ	STO	4	τ
3	Store variables				
	Initial value	X_0	STO	0	X_0
	Instantaneous value	X_t	STO	1	X_t
	Steady state value	X_{ss}	STO	2	X_{ss}
	Elapsed time	t	STO	3	t
	(Store any 3 of the 4 variables)				
4	To calculate:				
	X_0, initial value		GSB	0	X_0
	X_t, Instantaneous value		GSB	1	X_t
	X_{ss}, Steady state value		GSB	2	X_{ss}
	t, elapsed time		GSB	3	t
5	For a new case go to step 2				

Program Listings

Step	Key	Notes
01	*LBL1	X_t
02	RCL0	
03	RCL2	
04	-	
05	GSB5	
06	÷	
07	RCL2	
08	+	
09	R/S	***
10	*LBL2	X_{ss}
11	GSB5	
12	ENT↑	
13	ENT↑	
14	RCL1	
15	x	
16	RCL0	
17	-	
18	X⇄Y	
19	1	
20	-	
21	÷	
22	R/S	***
23	*LBL0	X_0
24	GSB5	
25	RCL1	
26	RCL2	
27	-	
28	x	
29	RCL2	
30	+	
31	R/S	***
32	*LBL3	t
33	RCL0	
34	RCL2	
35	-	
36	RCL1	
37	RCL2	
38	-	
39	÷	
40	LN	
41	RCL4	
42	x	
43	R/S	***
44	*LBL5	$e^{-t/\tau}$
45	RCL3	
46	RCL4	
47	÷	
48	e^x	
49	RTN	
50	R/S	

*** "Printx" may be inserted before "R/S".

REGISTERS

0 X_0	1 X_t	2 X_{ss}	3 t	4	5
6 τ	7	8	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

EQUATIONS OF MOTION

This program calculates an interchangeable solution among the variables: displacement, acceleration, initial velocity, and time or final velocity for an object that undergoes constant acceleration. The motion must be linear.

EQUATIONS:

Final velocity $v = \sqrt{v_0^2 + 2ax}$

Initial velocity $v_0 = \sqrt{v^2 - 2ax}$

Displacement $x = \dfrac{v^2 - v_0^2}{2a}$

Acceleration $a = \dfrac{v^2 - v_0^2}{2x}$

Displacement $x = v_0 t + \frac{1}{2} at^2$

Initial velocity $v_0 = \frac{x}{t} - \frac{1}{2} at$

Acceleration $a = \dfrac{x - v_0 t}{\frac{1}{2} t^2}$

Time $t = \dfrac{\sqrt{v_0^2 + 2ax} - v_0}{a}$

REMARKS:

Any consistent set of units may be used.

Displacement, acceleration, and velocity should be considered signed (vector) quantities. For example, if initial velocity and acceleration are in opposite directions, one should be positive and the other negative.

All equations assume initial displacement, x_0, equals zero.

EXAMPLE 1:

An automobile accelerates for 4 seconds from a speed of 35 mph and covers a distance of 264 feet. Assuming constant acceleration, what is the acceleration in ft/sec^2? (7.33 ft/sec^2) If the acceleration continues to be constant, what distance is covered in the next second? (84.33 ft)

SOLUTION:

```
 264.00 STO1   x
  35.00 ENT↑
5280.00   ×
3600.00   ÷
        STO2   v0
   4.00 STO3   t
        GSB4
   7.33  ***   a

   5.00 STO3   t + 1 sec
        GSB1
 348.33  ***   x at t + 1 sec

 264.00   -    x at t
  84.33  ***   x(t+1) - x(t)
```

EXAMPLE 2:

An airplane's take off velocity is 125 mph. Assume a constant acceleration of 15 ft/sec^2. How much runway length in feet will be used from start to take-off? (1120.37 ft.) How long will it take for the plane to reach take-off velocity? (12.22 sec)

SOLUTION:

```
   0.00 STO2   v0
  15.00 STO4
 125.00 ENT↑
5280.00    ×
3600.00    ÷
        STO5   v
        GSB7
1120.37  ***   x

        GSB3
  12.22  ***   t
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Store variables				
	displacement	x	STO	1	x
	initial velocity	v_0	STO	2	v_0
	time	t	STO	3	t
	acceleration	a	STO	4	a
	final velocity	v	STO	5	v
	(store appropriate unknowns)				
3.	To calculate:				
	A. Displacement (a, v_0 and t or v known)				
	if t is known		GSB	1	x
	if v is known		GSB	7	x
	B. Initial Velocity ($a, x,$ and t or v known)				
	if t is known		GSB	2	v_0
	if v is known		GSB	6	v_0
	C. Acceleration (v_0, x and t or v known)				
	if t is known		GSB	4	a
	if v is known		GSB	8	a
	D. Time (v_0, x and a known)		GSB	3	t
	E. Final Velocity (v_0, x and a known)		GSB	5	v
4.	For a new case go to step 2				

Program Listings

Program	Comments
01 *LBL1	x (t known)
02 RCL2	
03 RCL4	
04 2	
05 ÷	
06 RCL3	
07 x	
08 +	
09 RCL3	
10 x	
11 ST01	
12 R/S	***
13 *LBL2	v_0(t known)
14 RCL1	
15 RCL3	
16 ÷	
17 LSTX	
18 RCL4	
19 x	
20 2	
21 ÷	
22 -	
23 ST02	
24 R/S	***
25 *LBL4	a (t known)
26 RCL1	
27 RCL2	
28 RCL3	
29 x	
30 -	
31 RCL3	
32 X²	
33 2	
34 ÷	
35 ÷	
36 ST04	
37 R/S	***
38 *LBL3	t
39 GSB5	
40 RCL2	
41 -	
42 RCL4	
43 ÷	
44 ST03	
45 R/S	***
46 *LBL5	v
47 RCL2	
48 X²	
49 2	
50 GSB9	
51 ST05	
52 RTN	
53 *LBL9	
54 RCL4	
55 RCL1	
56 x	
57 x	
58 +	
59 √X	
60 RTN	
61 *LBL6	v_0(v known)
62 RCL5	
63 X²	
64 2	
65 CHS	
66 GSB9	
67 ST02	
68 R/S	***
69 *LBL7	x (v known)
70 4	
71 ST00	
72 GSB0	
73 ST01	
74 R/S	***
75 *LBL0	
76 RCL5	
77 X²	
78 RCL2	
79 X²	
80 -	
81 RCLi	
82 2	
83 x	
84 ÷	
85 RTN	
86 *LBL8	a (v known)
87 1	
88 ST00	
89 GSB0	
90 ST04	
91 R/S	***

*** "Printx" may be inserted before "R/S".

REGISTERS

0 Used	1 x	2 v_0	3 t	4 a	5 v
6	7	8	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

KINETIC ENERGY

This program calculates an interchangeable solution among the variables weight (or mass), velocity, and kinetic energy, for an object moving at constant velocity. The program operates in either English or metric units. For metric units, any consistent set of units may be used; the quantity mass must be used. For English units, the energy must be in foot-pounds, the velocity in feet per second, and the quantity weight in pounds.

K.E. = Kinetic energy

W = Weight (lb)

m = Mass (kg, g)

v = Velocity

g = Acceleration due to gravity = 32.17398 ft/sec^2

EQUATIONS:

English

$$K.E. = \frac{1W}{2g} v^2$$

Metric

$$K.E. = \frac{1}{2} m v^2$$

1 ft-lb = 1.98 x 10^6 hp

EXAMPLE 1:

The slider of a slider-crank mechanism is used to punch holes in a slab of metal. It is found that the work required to punch a hole is 775 ft-lb. If the slider weighs 5 lb. 4 oz., how fast must it be moving when it strikes the metal? (97.46 ft/sec) What is the required work in horsepower? (3.91 x 10^{-4} hp)

SOLUTION:

```
          GSB2
  64.35   ***    English Units

 775.00 STO1
   5.00 ENT↑
   4.00 ENT↑
  16.00   ÷
          +
   5.25   ***    W
        STO2

          GSB5
  97.46   ***    v

          GSB3
 775.00   ***    ft-lb
          R/S
3.91-04   ***    hp
```

EXAMPLE 2:

An object weighing 4.8 kg is moving with constant velocity of 3.5 m/sec. Find its kinetic energy. (29.40 joules)

SOLUTION:

```
          GSB1
   2.00   ***    Metric Units

   4.80 STO2
   3.50 STO3
          GSB3
  29.40   ***    K.E.
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1	Key in the program				
2	Choose system of units				
	Metric (SI)		GSB	1	2.00
	or				
	English		GSB	2	64.35
3	Input two of the following variables				
	Kinetic energy	K.E.	STO	1	K.E.
	Weight (mass)	W(m)	STO	2	W(m)
	Velocity	v	STO	3	v
4	Compute the remaining variables:				
	Kinetic energy		GSB	3	K.E.
	Optional: convert K.E.(ft-lb) to K.E.(hp)		R/S		K.E.(hp)
	Weight (mass)		GSB	4	W(m)
	Velocity		GSB	5	v
5	To change any input variable go to step 3				
6	For a new case, go to step 2				

Program Listings

01 *LBL1	Metric
02 2	
03 STO4	
04 R/S	
05 *LBL2	English
06 6	
07 4	
08 .	
09 3	
10 4	
11 7	
12 9	
13 6	
14 STO4	
15 R/S	
16 *LBL3	K.E. Calc.
17 RCL2	
18 RCL3	
19 ENT↑	
20 x	
21 x	
22 RCL4	
23 ÷	
24 STO1	***
25 R/S	
26 RCL1	
27 1	
28 9	
29 8	
30 EEX	
31 4	
32 ÷	***
33 R/S	
34 *LBL4	W(m) Calc.
35 RCL1	
36 RCL4	
37 x	
38 RCL3	
39 ENT↑	
40 x	
41 ÷	
42 STO2	***
43 R/S	
44 *LBL5	v Calc.
45 RCL1	
46 RCL4	
47 x	
48 RCL2	
49 ÷	
50 √X	
51 STO3	***
52 R/S	

*** "Printx" may be inserted before "R/S".

REGISTERS

0	1 K.E.	2 W(m)	3 v	4 2(met) 64.3(Eng)	
6	7	8	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

RPM/TORQUE/POWER

This program provides an interchangeable solution for RPM, torque, and power in both Systeme International (metric) and English units.

	SI	English
RPM	RPM	RPM
Torque	nt-m	ft-lb
Power	watts	hp

EQUATIONS:

RPM x Torque = Power

1 hp = 745.7 watts

1 ft-lb = 1.356 joules

1 RPM = $\pi/30$ radians/sec

1 hp = 550 $\frac{\text{ft-lb}}{\text{sec}}$

EXAMPLE 1:

Calculate the torque from an engine developing 11 hp at 6500 RPM. Find the SI equivalent.

EXAMPLE 2:

A generator is turning at 1600 RPM with a torque of 20 nt-m. If it is 90% efficient, what is the power input in both systems?

SOLUTIONS:

(1)

	GSB4	
6500.00	ENT↑	
0.00	ENT↑	
11.00	GSB5	
8.89	***	Torque, ft-lb
	R/S	
12.05	***	Torque, nt-m

(2)

	GSB3	
1600.00	ENT↑	
20.00	ENT↑	
0.90	÷	
0.00	GSB5	
3723.37	***	Power, watts
	R/S	
4.99	***	Power, hp

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Choose sytem of units:				
	Metric		GSB	3	1.36
	or				
	English		GSB	4	5251.41
3.	Enter variables (the unknown quantity, x,				
	must be input as zero) and compute x.				
	RPM	RPM	ENT↑		
	Torque	Torque	ENT↑		
	Power	Power	GSB	5	x
4.	(Optional) to convert torque or power to				
	other system		R/S		x converted

Program Listings

01	*LBL3	
02	3	Set up for metric units
03	0	
04	Pi	
05	÷	
06	STO7	
07	7	
08	4	
09	5	
10	.	
11	7	
12	STO5	
13	1	
14	.	
15	3	
16	5	
17	6	
18	STO6	
19	RTN	
20	*LBL4	Set up for English units
21	GSB3	
22	1/X	
23	STO6	
24	X⇄Y	
25	1/X	
26	STO5	
27	÷	
28	x	
29	STO7	
30	RTN	
31	*LBL5	
32	4	
33	STO0	
34	R↓	
35	*LBL8	
36	STOi	Store variables
37	R↓	
38	DSZ	
39	GTO8	
40	*LBL9	
41	X=0?	Determine quantity to calculate
42	GTOi	
43	ISZ	
44	R↓	
45	GTO9	
46	*LBL2	
47	RCL4	
48	RCL3	
49	÷	
50	RCL7	
51	x	
52	R/S	** RPM
53	*LBL1	
54	RCL4	
55	RCL2	
56	÷	
57	RCL7	
58	x	
59	R/S	*** Torque
60	RCL6	
61	÷	
62	R/S	** Torque converted
63	*LBL0	
64	RCL2	
65	RCL3	
66	x	
67	RCL7	
68	÷	
69	R/S	*** Power
70	RCL5	
71	÷	
72	R/S	** Power converted

** "Printx" may be inserted before "R/S".
*** "Printx" may replace "R/S".

REGISTERS

0 i	1 Used	2 RPM	3 Torque	4 Power	5 Used
6 Used	7 Used	8	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

BLACK BODY THERMAL RADIATION

Bodies with finite temperatures emit thermal radiation. The higher the absolute temperature, the more thermal radiation emitted. Bodies which emit the maximum possible amount of energy at every wavelength for a specified temperature are said to be black bodies. While black bodies do not actually exist in nature, many surfaces may be assumed to be black for engineering considerations.



Figure 1.

Figure 1 is a representation of black body thermal emission as a function of wavelength. Note that as temperature increases the area under the curves (total emissive power $E_{b(0-\infty)}$) increases. Also note that the wavelength of maximum emissive power λ_{max} shifts to the left as temperature increases.

This program can be used to calculate the wavelength of maximum emissive power for a given temperature, the temperature corresponding to a particular wavelength of maximum emissive power, the total emissive power for all wavelengths and the emissive power at a particular wavelength. It can also be used to calculate the emissive power from zero to an arbitrary wavelength, the emissive power between two wavelengths or the total emissive power.

EQUATIONS:

$$\lambda_{max}\ T_{\lambda_{max}} = c_3$$

$$E_{b(0-\infty)} = \sigma T^4$$

$$E_{b\lambda} = \frac{2\pi c_1}{\lambda^5(e^{c_2/\lambda T}-1)}$$

$$E_{b(0-\lambda)} = \int_0^{\lambda} E_{b\lambda}\ d\lambda$$

$$= 2\pi c_1 \sum_{k=1}^{\infty} -T/kc_2\ e^{-\frac{kc_2}{T\lambda}} \left[\left(\frac{1}{\lambda}\right)^3 + \frac{3T}{\lambda^2 kc_2} + \frac{6}{\lambda}\left(\frac{T}{kc_2}\right)^2 + 6\left(\frac{T}{kc_2}\right)^3\right]$$

$$E_{b(\lambda_1-\lambda_2)} = E_{b(0-\lambda_2)} - E_{b(0-\lambda_1)}$$

where

λ_{max} is the wavelength of maximum emissivity in microns;

T is the absolute temperature in °R or K;

$E_{b(0-\infty)}$ is the total emissive power in Btu/hr-ft^2 or Watts/cm^2;

$E_{b\lambda}$ is the emissive power at λ in Btu/hr-ft^2-μm or Watts/cm^2-μm;

$E_{b(0-\lambda)}$ is the emissive power for wavelengths less than λ in Btu/hr-ft^2 or Watts/cm^2;

$E_{b(\lambda_1-\lambda_2)}$ is the emissive power for wavelengths between λ_1 and λ_2 in Btu/hr-ft^2 or Watts/cm^2.

c_1 = 1.8887982 x 10^7 Btu-μm^4/hr-ft^2
= 5.9544 x 10^3 Wμm^4/cm^2

c_2 = 2.58984 x 10^4 μm-°R =
1.4388 x 10^4 μm-K

c_3 = 5.216 x 10^3 μm-°R =
2.8978 x 10^3 μm-K

σ = 1.71312 x 10^{-9} Btu/hr-ft^2-°R^4=
5.6693 x 10^{-12} W/cm^2-K^4

σ_{exp} = 1.731 x 10^{-9} Btu/hr-ft^2-°R^4
= 5.729 x 10^{-12} W/cm^2-K^4

REMARKS:

A minute or more may be required to obtain $E_{b(0-\lambda)}$ or $E_{b(\lambda_1-\lambda_2)}$ since the integration is numerical.

Sources differ on values for constants. This could yield small discrepancies between published tables and outputs.

REFERENCE:

Robert Siegel and John R. Howell, Thermal Radiation Heat Transfer, Vol. 1, National Aeronautics and Space Administration, 1968.

EXAMPLE 1:

What percentage of the radiant output of a lamp is in the visible range (0.4 to 0.7 microns) if the filament of the lamp is assumed to be a black body at 2400 K?

EXAMPLE 2:

If the human eye was designed to work most efficiently in sunlight and the visible spectrum runs from about 0.4 to 0.7 microns, what is the sun's temperature in degrees Rankine? Assume that the sun is a black body. Using the temperature calculated, find the fraction of the sun's total emissive power which falls in the visible range. Find the percentage of the sun's radiation which has a wavelength less than 0.4 microns.

SOLUTIONS:

1.

```
   5954.40 ST01  }
  14388.00 ST02  } S.I. constants
   2897.80 ST03  }
5.6693-12  ST04  }
   2400.00 ST05
      0.40 ST06
      0.70 ST07
           GSB4
      4.97  ***
           GSB2    Eb (0 to ∞)
              ÷
    100.00    ×
      2.64  ***    (%)
```

2.

18887982.00	STO1	English constants
25898.40	STO2	
5216.00	STO3	
1.71312-09	STO4	
0.40	ENT↑	
0.70	+	
2.00	÷	
0.55	***	mean value
	RCL3	
	÷	
	1/X	
9483.64	***	T, (°R)
	STO5	
0.40	STO6	
0.70	STO7	
	GSB4	
4670556.56	***	E_b(.4 to .7)
	GSB2	
13857578.83	***	E_b(0 to ∞)
	÷	
100.00	×	
33.70	***	(%)
0.40	STO6	
	GSB1	
1168606.94	***	E_b (0 to .4)
	GSB2	
	÷	
100.00	×	
8.43	***	(%)

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1	Key in the program				
2	Store constants:				
2a	English units -	18887982	STO	1	
	(Btu, μm, ft, °R)	25898.4	STO	2	
		5216	STO	3	
	or	$.171312 \times 10^{-8}$	STO	4	
2b	SI units -	5954.4	STO	1	
	(W, μm, cm, °K)	14388	STO	2	
		2897.8	STO	3	
		5.6693×10^{-12}	STO	4	
3	For experimental Stefan-Boltzmann constant	1.0105	STO	x	
	instead of theoretical constant		4		
4	To calculate $\lambda_{max} = f(T)$		RCL	3	
		T(°R or k)	÷		λ_{max}(μm)
5	To calculate T = f(λ) for which λ is		RCL	3	
	maximum	λ(μm)	÷		T(°R or k)
6	To calculate total emissive power	T*	STO	5	
			GSB	2	$E_b(0 \text{ to } \infty)$
7	To calculate emissive power at λ	T*	STO	5	
		λ	STO	6	
			GSB	3	$E_b(\lambda)$
8	To calculate emissive power from 0 to λ_1	T*	STO	5	
		λ_1	STO	6	
			GSB	1	$E_b(0 \text{ to } \lambda_1)$
9	To calculate emissive power from λ_1 to λ_2	T*	STO	5	
		λ_1	STO	6	
	*any value of T stored previously is still	λ_2	STO	7	
	stored and need not be input again		GSB	4	$E_b(\lambda_1 \text{ to } \lambda_2)$

Program Listings

01	*LBL1	
02	GSB9	
03	R/S	***$E_b(0 \text{ to } \lambda_1)$
04	*LBL9	
05	0	
06	ST09	
07	ST08	
08	*LBL0	
09	RCL2	
10	RCL5	
11	÷	
12	ST-8	$-k\ c_2/T$
13	3	
14	RCL8	
15	÷	
16	RCL6	
17	X²	
18	÷	
19	LSTX	λ^2
20	RCL6	
21	x	
22	1/X	
23	-	
24	6	
25	RCL6	
26	÷	
27	RCL8	
28	X²	
29	÷	
30	-	
31	6	
32	RCL8	
33	3	
34	Yˣ	
35	÷	
36	+	
37	RCL8	
38	RCL6	
39	÷	
40	eˣ	
41	x	
42	RCL8	
43	÷	
44	ST+9	Δ
45	RCL9	
46	÷	
47	EEX	
48	CHS	
49	5	
50	X≤Y?	$\Delta \geq .001\%$
51	GTO0	yes, increment k
52	RCL9	
53	2	
54	x	
55	Pi	
56	x	
57	RCL1	
58	x	
59	RTN	$E_b(0 \text{ to } \lambda)$
60	*LBL2	
61	RCL5	
62	4	
63	Yˣ	
64	RCL4	
65	x	
66	R/S	***$E_b(0 \text{ to } \infty)$
67	*LBL3	
68	RCL1	
69	2	
70	x	
71	Pi	
72	x	
73	RCL6	
74	5	
75	Yˣ	
76	÷	
77	RCL2	
78	RCL6	
79	÷	
80	RCL5	
81	÷	
82	eˣ	
83	1	
84	-	
85	÷	
86	R/S	***$E_{b\lambda}$
87	*LBL4	
88	GSB9	
89	ST.0	$E_b(0 \text{ to } \lambda_1)$
90	RCL7	
91	ST06	λ_2
92	GSB9	
93	RC.0	
94	-	
95	R/S	***$E_b(\lambda_1 \text{ to } \lambda_2)$

***"Printx" may be inserted before "R/S"

REGISTERS

0	1 c_1	2 c_2	3 c_3	4 σ	5 T
6 λ	7 λ_2	8 $-Kc_2/T$	9 sum	.0 used	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

CONSERVATION OF ENERGY

This program converts kinetic energy, potential energy, and pressure-volume work to energy. Energy is stored as a running total which may at any time be converted to an equivalent velocity, height, pressure, or energy per unit mass. The program is useful in fluid flow problems where velocity, elevation and pressure change along the path of flow.

EQUATIONS:

$$\frac{v_1^2}{2} + gz_1 + \frac{P_1}{\rho} + \frac{E_1}{\dot{m}} =$$

$$\frac{v_2^2}{2} + gz_2 + \frac{P_2}{\rho} + \frac{E_2}{\dot{m}}$$

where:

- v is the fluid velocity;
- z is the height above a reference datum;
- P is the pressure;
- E is an energy term which could represent inputs of work or friction loses (negative value);
- g is the acceleration of gravity;
- ρ is the fluid density;
- $\dot{m}$ is the mass flow rate (assumed to be unity);

  subscripts 1 and 2 refer to upstream and downstream values respectively.

NOTES:

Downstream values should be input as negatives. However, when an output is called for, the calculator displays the relative value with no regard to upstream or downstream location.

An error will result when the total energy sum stored in register 8 is negative and an attempt is made to calculate velocity.

EXAMPLE 1:

A water tower is 100 feet high. What is the zero flow rate pressure at the base? The density of water is 62.4 lb/ft^3.

If water is flowing out of the tower at a velocity of 10 ft/sec, what is the static pressure?

What is the maximum frictionless flow velocity which could be achieved with the 100 foot tower?

If 10000 pounds of water are pumped to the top of the tower every hour, at a velocity of 20 ft/sec, with a frictional pressure drop of 2 psi, how much power is needed at the pump?

EXAMPLE 2:

An incompressible fluid (ρ = 735 kg/m) flows through the converging passage of Figure 1. At point 1 the velocity is 3 m/s and at point 2 the velocity is 15 m/s. The elevation difference between points 1 and 2 is 3.7 meters. Assuming frictionless flow, what is the static pressure difference between points 1 and 2?



Figure 1.

EXAMPLE 3:

A reservoir's level is 25 meters above the discharge pond. Assuming 85% power generation efficiency, how much power can be generated with a flow rate of 20 m^3/s?

$\rho = 1000\ kg/m^3$

SOLUTIONS:

(1)

```
25033.407 ST05
    32.17 ST06
  4632.48 ST07
    62.40 GSB1
   100.00 GSB3
          GSB8
    43.33  ***   (psig)
   -10.00 GSB2
          GSB8
    42.66  ***   (psig)
    62.40 GSB1
   100.00 GSB3
          GSB6
    80.21  ***   (ft/sec)
    62.40 GSB1
    20.00 GSB2
     2.00 GSB4
   100.00 GSB3
          GSB9
     0.14  ***   (BTU/lb)
 10000.00   x
  1424.29  ***   (BTU/hr)
```

(2)

```
      1.00 ST05
           ST07
   9.80665 ST06
    735.00 GSB1
      3.00 GSB2
      3.70 GSB3
    -15.00 GSB2
           GSB8
 -52710.82  ***   (Nt/m²)
```

(3)

```
   1000.00 GSB1
     25.00 GSB3
           GSB9
    245.17  ***   (joule/kg)
      0.85   x
    208.39  ***   (joule/kg)
     20.00 ENT↑
   1000.00   x    (kg/s)
             x
4167826.25  ***   (watts)
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	For English units:	25033.407	STO	5	
		32.17	STO	6	
		4632.48	STO	7	1.00
	For S.I. units:	9.80665	STO	6	
3.	Enter fluid density	ρ	GSB	1	0.00
4.	Enter the following (negative values are				
	downstream values):				
	Fluid velocity	v	GSB	2	
	Height from reference datum	z	GSB	3	
	Pressure	P	GSB	4	
	Energy input	E	GSB	5	
5.	Repeat step 4 for all input values				
6.	Calculate the unknown:				
	Fluid velocity		GSB	6	v
	Height from reference datum		GSB	7	z
	Pressure		GSB	8	P
	Energy		GSB	9	E
7.	For another case, go to step 3, or clear				
	register 8 and go to step 4	0	STO	8	

Program Listings

01	*LBL1	
02	STO4	ρ
03	0	
04	STO8	Clear Σ E
05	R/S	
06	*LBL2	
07	ENT↑	
08	ABS	
09	x	
10	2	
11	÷	$\pm\ v^2/_2$
12	GTO5	
13	*LBL3	
14	RCL6	
15	x	
16	GTO5	gz
17	*LBL4	
18	RCL7	
19	x	
20	RCL4	
21	÷	$^{p}/\rho$
22	*LBL5	E
23	ST+8	
24	R/S	
25	*LBL6	
26	RCL8	
27	2	
28	x	
29	√X	
30	R/S	*** v
31	*LBL7	
32	RCL8	
33	RCL6	
34	÷	
35	R/S	*** z
36	*LBL8	
37	RCL8	
38	RCL7	
39	÷	
40	RCL4	
41	x	
42	R/S	*** P
43	*LBL9	
44	RCL8	
45	RCL5	
46	÷	
47	R/S	*** E

*** "Printx" may be inserted before "R/S".

REGISTERS

0	1	2	3	4 ρ	5 Used
6 g	7 Used	8 Σ E	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

MOHR CIRCLE FOR STRESS

Given the state of stress on an element, the principal stresses and their orientation can be found. The maximum shear stress and its orientation can also be found.



EQUATIONS:

$$s_{smax} = \sqrt{\left(\frac{s_x - s_y}{2}\right)^2 + s_{xy}^2}$$

$$s_1 = \frac{s_x + s_y}{2} + s_{smax}$$

$$s_2 = \frac{s_x + s_y}{2} - s_{smax}$$

$$\theta = 1/2 \tan^{-1}\left(\frac{2s_{xy}}{s_x - s_y}\right)$$

$$\theta_s = 1/2 \tan^{-1} - \left(\frac{s_x - s_y}{2s_{xy}}\right)$$

where:

s_{smax} is the maximum shear stress;

s_1 and s_2 are the principal normal stresses;

θ is the angle of rotation from the principal axis to the original axis;

θ_s is the angle of rotation from the axis of maximum shear stress to the original axis;

s_x is the stress in the x direction;

s_y is the stress in the y direction;

s_{xy} is the shear stress on the element.

REFERENCE:

Spotts, M.F., Design of Machine Elements, Prentice-Hall, 1971.

EXAMPLE:

If s_x = 25000 psi, s_y = -5000 psi, and s_{xy} = 4000 psi, compute the principal stresses and the maximum shear stress.



SOLUTION:

```
25000.00 ENT↑
-5000.00 ENT↑
 4000.00 GSB1
25524.17  ***
          R/S   s1 (psi)
-5524.17  ***   s2 (psi)
          R/S
    7.47  ***   θ (degrees)
          R/S
  -37.53  ***   θs (degrees)
          R/S
15524.17  ***   ssmax (psi)
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Enter the following:				
	Stress in the x direction (negative for compression)	S_x	ENT↑		
	Stress in the y direction (negative for compression)	S_y	ENT↑		
	Shear stress	S_{xy}			
3.	Compute the following:				
	First principal stress		GSB	1	S_1
	Second principal stress		R/S		S_2
	Angle of rotation (principal)		R/S		θ
	Angle of rotation (shear)		R/S		θ_s
	Maximum shear stress		R/S		s_{smax}
	NOTE: Do not disturb the stack during step 3				
4.	For a new case, go to step 2.				

Program Listings

01	*LBL1		
02	ENT↑		
03	R↓	s_{xy} s_y s_x s_{xy}	
04	STO3		
05	R↓		
06	X⇄Y		
07	STO1	s_x	
08	X⇄Y	s_y	
09	ST+1		
10	-	$s_x - s_y$	
11	2		
12	ST÷1		
13	÷		
14	STO4	$(s_x - s_y)/2$	
15	→P		
16	STO2	s_{smax}	
17	RCL1		
18	+		
19	R/S	** s_1	
20	X⇄Y	$2\cdot\theta$	
21	RCL1		
22	RCL2		
23	-		
24	R/S	** s_2	
25	X⇄Y		
26	2		
27	÷		
28	R/S	** θ	
29	RCL4		
30	RCL3		
31	÷		
32	CHS		
33	TAN⁻¹		
34	2		
35	÷		
36	R/S	** θ_s	** "Printx" may replace "R/S".
37	RCL2		
38	R/S	*** s_{smax}	*** "Printx" may be inserted before "R/S".

REGISTERS

0	1 $(s_x+s_y)/2$	2 s_{smax}	3 s_{xy}	4 $(s_x-s_y)/2$	5
6	7	8	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

POLYNOMIAL EVALUATION--REAL OR COMPLEX

This program evaluates polynomials of the form:

$f(x_0) = a_0+a_1x+...+a_nx^n$

where the coefficients

$a_k, k=0,...n (n \leq 28)$ and x_0 are real or the coefficients and x_0 are complex of the form

$a_k = \text{Re}(a_k) + i \text{ Im}(a_k)$

$z_0 = \text{Re}(z_0) + i \text{ Im}(z_0)$

$k = 0, 1, ... n$

Example 1:

$f(x) = 11-7x - 3x^2 + 5x^4 + x^5$

for $x_0 = 2.5$

for $x_0 = -5$

Solution:

```
         CLRG
 11.00   GSB1
 -7.00   R/S
 -3.00   R/S
  0.00   R/S
  5.00   R/S
  1.00   R/S
  2.50   GSB2
267.72   ***     f (2.5)
  6.00   STO0
 -5.00   GSB2
-29.00   ***     f (-5)
```

Example 2:

$f(x) = (5-7i) - 10x + (-2+i)x^2$
$+ 18x^3 + (3+4i)x^4$

for $x_0 = 2 + i$

Solution:

```
   1.00  ENT↑
   2.00  GSB3
   4.00  ENT↑
   3.00  GSB4
   0.00  ENT↑
  18.00  GSB4
   1.00  ENT↑
  -2.00  GSB4
   0.00  ENT↑
 -10.00  GSB4
  -7.00  ENT↑
   5.00  GSB5
-106.00  ***     Re f(x0)
         R/S
 220.00  ***     Im f(x0)
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
	For real polynomials:				
2.	Initialize		f	REG	
3.	Store coefficients	a_0	GSB	1	
4.	Continue for i = 1 to n, n≤28	a_i	R/S		
5.	Compute $f(x_0)$	x_0	GSB	2	$f(x_0)$
6.	For a different x_0 go to step 5	n+1	STO	0	
	For complex polynomials:				
7.	Enter x_0	Im x_0	ENT↑		
		Re x_0	GSB	3	0
8.	Enter a_k, k=n,n-1,...1	Im a_k	ENT↑		
		Re a_k	GSB	4	
9.	Enter a_0 and run	Im a_0	ENT↑		
		Re a_0	GSB	5	Re $f(x_0)$
			R/S		Im $f(x_0)$
10.	For a new case, go to step 7				

Step	Key	Comment
01	*LBL1	
02	ISZ	
03	STOi	
04	R/S	
05	GTO1	
06	*LBL2	
07	ENT↑	
08	ENT↑	
09	ENT↑	
10	RCLi	
11	×	Multiply by x_0
12	DSZ	
13	*LBL0	
14	RCLi	
15	+	
16	×	Multiply by x_0
17	DSZ	
18	GTO0	
19	X⇄Y	
20	÷	Divide by x_0
21	R/S	*** $f(x_0)$
22	*LBL3	Routines for
23	→P	complex polynomials
24	STO1	r
25	X⇄Y	
26	STO2	θ
27	0	
28	ENT↑	Prepare for LBL 9
29	ENT↑	
30	ENT↑	
31	RTN	
32	*LBL4	
33	GSB9	
34	GTO8	
35	*LBL5	
36	GSB9	
37	R/S	*** Re $f(x_0)$
38	X⇄Y	
39	R/S	*** Im $f(x_0)$
40	*LBL8	Multiply in polar form
41	→P	
42	RCL1	
43	×	
44	X⇄Y	
45	RCL2	
46	+	
47	X⇄Y	
48	→R	
49	RTN	
50	*LBL9	Add real parts and imaginary parts
51	X⇄Y	
52	R↓	
53	+	
54	R↓	
55	+	
56	X⇄Y	
57	R↓	
58	X⇄Y	
59	RTN	
60	R/S	

REGISTERS

0 i	1 r or a_0	2 θ or a_1	3 a_2	4 . . . a_n	5
6	7	8	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29 a_{28}

*** "Printx" may be inserted or used to replace "R/S".

SINE, COSINE, AND EXPONENTIAL INTEGRALS

Sine integral:

$$Si(x) = \int_0^x \frac{\sin t}{t}\, dt = \sum_{n=0}^{\infty} \frac{(-1)^n x^{2n+1}}{(2n+1)(2n+1)!}$$

where x is real.

This routine computes successive partial sums of the series, stops when two consecutive partial sums are equal, and displays the last partial sum as the answer.



Notes: When x is too large, computing a new term of the series might cause an overflow. In that case, display shows all 9's and the program stops.

$Si(-x) = -Si(x)$

Cosine integral:

$$Ci(x) = \gamma + \ln x + \int_0^x \frac{\cos t - 1}{t}\, dt = \gamma + \ln x + \sum_{n=1}^{\infty} \frac{(-1)^n x^{2n}}{2n(2n)!}$$

where $x>0$, and $\gamma = 0.577215665$ is the Euler's constant.

This program computes successive partial sums of the series. When two consecutive partial sums are equal, the value is used as the sum of the series.



Notes: When x is too large, computing a new term of the series might cause an overflow. In that case, display shows all 9's and the program stops.

$Ci(-x) = Ci(x) - i\pi$ for $x>0$.

Exponential integral:

$$Ei(x) = \int_{-\infty}^{x} \frac{e^t}{t}\, dt = \gamma + \ln x + \sum_{n=1}^{\infty} \frac{x^n}{n\,n!}$$

where $x>0$ and $\gamma = 0.577215665$ is Euler's constant.

This program computes successive partial sums of the series. When two consecutive partial sums are equal, the value is used as the sum of the series.



Note: When x is too large, computing a new term of the series might cause an overflow. In that case, display shows all 9's and the program stops.

References: Handbook of Mathematical Functions, Abramowitz and Stegun, National Bureau of Standards, 1968.

Examples:

1. Si (.69)
2. Si (.98)
3. Ci (1.38)
4. Ci (5)
5. Ei (1.59)
6. Ei (.61)

Solutions:

```
0.577215665 ST.0
       0.69 GSB1
       0.67  ***
       0.98 GSB1
       0.93  ***
       1.38 GSB2
       0.46  ***
       5.00 GSB2
      -0.19  ***
       1.59 GSB3
       3.57  ***
       0.61 GSB3
       0.80  ***
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Store α	.577215665	STO	•	
			0		α
3.	For Sine Integral	x	GSB	1	Si(x)
	For Cosine Integral	x	GSB	2	Ci(x)
	For Exponential Integral	x	GSB	3	Ei(x)

01 *LBL1	Sine Integral routine	50 RC.0	
02 STO3		51 +	
03 X²		52 *LBL9	
04 CHS		53 RCL1	
05 STO1		54 RCL2	
06 1		55 1	
07 STO2		56 +	
08 RCL3		57 STO2	
09 *LBL0		58 ÷	
10 RCL1		59 RCL3	
11 RCL2		60 x	
12 1		61 STO3	
13 +		62 RCL2	
14 ÷		63 ÷	
15 LSTX		64 +	
16 1		65 X≠Y?	
17 +		66 GTO9	
18 STO2		67 R/S	*** Ei(x)
19 ÷			
20 RCL3			
21 x			
22 STO3			
23 RCL2			
24 ÷			
25 +			
26 X≠Y?			
27 GTO0			
28 R/S	*** Si(x)/Ci(x)		
29 *LBL2	Cosine Integral routine		
30 X²			
31 CHS			
32 STO1			
33 1			
34 STO3			
35 0			
36 STO2			
37 LSTX			
38 LN			
39 RC.0			
40 +			
41 GTO0			
42 *LBL3			
43 STO1	Exponential Integral routine		
44 1			
45 STO3			
46 0			
47 STO2			
48 RCL1			
49 LN			

REGISTERS

0	1 used	2 used	3 used	4	5
6	7	8	9	.0 used	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

*** "Print X" may be used to replace "R/S".

NOTES

In the Hewlett-Packard tradition of supporting HP programmable calculators with quality software, the following titles have been carefully selected to offer useful solutions to many of the most often encountered problems in your field of interest. These ready-made programs are provided with convenient instructions that will allow flexibility of use and efficient operation. We hope that these Solutions books will save your valuable time. They provide you with a tool that will multiply the power of your HP-19C or HP-29C many times over in the months or years ahead.

Mathematics Solutions
Statistics Solutions
Financial Solutions
Electrical Engineering Solutions
Surveying Solutions
Games
Navigational Solutions
Civil Engineering Solutions
Mechanical Engineering Solutions
Student Engineering Solutions

HEWLETT PACKARD

Reorder No. 00029-14010 Printed in U.S.A. 00029-90014